Arbeitswerttheorie nach Marx
Ein dezentraler Ansatz

Hans-Gert Gräbe, Universität Leipzig
http://www.hg-graebe.de

Vortrag auf der Tagung
Kybernetik - evolutionäre Systemtheorie - Dialektik,
Berlin, 7.-10.11.2007


```
export(linalg):
DM:=Dom::Matrix():
Dg:=proc(u) begin Dom::SquareMatrix(nops(u))([op(u)],Diagonal) end:
Id:=proc(n) begin Dg([1$i=1..n]) end_proc:
```

Die Motivation

Die folgenden Überlegungen entspringen dem Versuch einer Rekonstruktion wesentlicher Momente der Marxschen Arbeitswerttheorie unter Berücksichtigung der Annahme, dass in einem entwickelten, durch den umfassenden Einsatz von Wissenschaft und Technik geprägten Arbeitsumfeld

> *die Schöpfung des wirklichen Reichtums weniger abhängt von der Arbeitszeit und dem Quantum angewandter Arbeit als von der Macht der Agentien, die während der Arbeitszeit in Bewegung gesetzt werden und die selbst wieder [...] in keinem Verhältnis steht zur unmittelbaren Arbeitszeit, die ihre Produktion kostet, sondern vielmehr abhängt vom allgemeinen Stand der Wissenschaft und dem Fortschritt der Technologie.* (MEW 42, S. 592)

Die Denkrichtung

"Creators and Owners'' - die zunehmende Bedeutung dezentraler unternehmerischer Entscheidungen ist adäquat zu berücksichtigen.

Grundlegende Begriffe der Marxschen ökonomischen Theorie sind mit dem dezentralen Ansatz der Input-Output-Analyse zu verbinden.

Die auf eine Lohnarbeiterperspektive zentrierten Semantiken der Marxschen Arbeitswerttheorie sind aufzubrechen.

Die Überlegungen von Peter Ruben (1995, 1998) sind in ein umfassenderes mathematisches Konzept einzubetten.

Ausgangspunkt und weiterer Auslöser der Überlegungen war ein Aufsatz von Peter Fleissner zum Transformationsproblem.

Die wichtigsten Thesen vorab

(1) *Wert* wird als quantitatives Maß nicht für Arbeit schlechthin, sondern für Arbeit auf ein fremdes Bedürfnis hin gefasst, die Kompensation in "gleicher Höhe" durch Befriedigung eigenen Bedürfnisses finden muss.

(2) *Wert als gesellschaftliches Verhältnis* findet auch quantitativ seinen Ausdruck in den individuellen Wertvorstellungen der einzelnen Produzenten. Auf dem Markt treffen sich damit nicht "Produkte voneinander unabhängig betriebner Privatarbeiten" (MEW23, S. 87), sondern gesellschaftliche Produzenten.

Diese Wertvorstellungen haben ihre Quelle in deren praktischen Erfahrungen, sind aber kommunikativ und mimetisch eng aufeinander bezogen und konstituieren so einen gesellschaftlichen Kohärenzprozess. Diese Wertvorstellungen sind eine wichtige, aber keineswegs die einzige Grundlage für Entscheidungen dieser Produzenten am Markt.

(3) Allein als dieser Kohärenzprozess und nicht als Hegelsche Wesenskategorie ließe sich der Begriff *Wertsubstanz* im Singular sinnvoll fassen. Ich verwende den Begriff jedoch aus prinzipiellen epistemologischen Erwägungen nicht in diesem Sinne, sondern nur im Zusammenhang mit subjektiven Wertrechnungen einzelner Produzenten.

(4) Wert tritt in seinen konkreten Formen (Plural !) nicht nur über *Preise* in Erscheinung, sondern auch über *individuelle Wertsubstanzrechnungen*, die zusammen ein dezentrales Buchhaltungssystem (die Konten der einzelnen Marktteilnehmer) konstituieren. Zahlungen am Markt (die "Realisierung des Werts") sind Teil eines gesellschaftlich kodierten Legitimationsprozesses dieser individuellen Wertsubstanzrechnungen.

(5) Realisierter Wert und Preis fallen in diesem Sinne an der Basis, den Zahlungen am Markt, zusammen.

Das "Auseinanderfallen von Wert und Preis" wird konsequent als Auseinanderfallen von individuellen Wertvorstellungen und realisiertem Wert interpretiert, welches seinerseits erst im Studium komplexerer Zusammenhänge auf verschiedenen Skalen verständlich wird. Diese Mehrskaleneffekte lassen sich nicht sinnvoll in einer einfachen Dichotomie von Wert als Wesen und Preis als Erscheinung auffangen.

(6) Arbeit auf fremdes Bedürfnis erschöpft sich nicht in einfacher produktiver Arbeit, wie sie etwa in (MEW23, S. 66) durchschimmert, sondern auch der (produktiv tätige) Kapitalist, der "mit schlauem Kennerblick die für sein besondres Geschäft ... passenden Produktionsmittel und Arbeitskräfte auswählt" (MEW23, S. 199), ist auf fremdes Bedürfnis hin tätig.

Im Gegensatz zu Marx, insb. (MEW 23, S. 207/208) gehe ich davon aus, dass auch unternehmerische Tätigkeit Quelle von Wert ist. Es stellt sich heraus, dass in einer solchen Rechnung der Wertanteil, welchen unternehmerische Tätigkeit einträgt, gerade das ist, was Marx als Mehrwert bezeichnet.

(7) Andererseits gehe ich mit (Ruben-98) davon aus, dass Lohnarbeit keine Ware ist, sondern ein Verdingungsverhältnis, und somit die Grundannahme in (MEW23, S. 200) nicht zu halten ist. Dazu wird auch für Lohnarbeiter eine (rudimentäre) individuelle Wertsubstanzrechnung geführt, die auf der mathematischen Seite zeigt, warum man Lohnarbeit trotzdem *wie* eine Ware rechnen kann.

Neben dem Dimensionsargument von Ruben wende ich vor allem ein, dass es ein grundlegender Unterschied ist, einen Menschen oder ein Pferd für einen Tag zu mieten (MEW 23, S. 200), da ersteres ein Verhältnis *innerhalb* der menschlichen Gattung ist und damit *innerhalb* des Wertverhältnisses zu behandeln und nicht wie eine Ware als Target, *auf dem* sich das Wertverhältnis ausdrückt im Sinne von (MEW 23, S. 52 ff.).

Das Modell

Dazu mündliche Ausführungen, die im zweiten Teil an einem
Beispiel aus einer Arbeit von Peter Fleissner und dessen Variationen
erläutert werden.

Fleissners Beispiel

Die Güterbilanz

Die Matrix der Produktionskoeffizienten

```
A:=DM([[1/10,0],[2,3/10]]):float(A);
```

$$\begin{pmatrix} 0.1 & 0 \\ 2.0 & 0.3 \end{pmatrix}$$

Der Produktionsausstoß und Ressourceneinsatz nach Sektoren

```
x:=DM([10,100]);
A*Dg(x); A*x;
```

$$\begin{pmatrix} 10 \\ 100 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 \\ 20 & 30 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 50 \end{pmatrix}$$

Die Konsumtion, unterteilt nach privatem Verbrauch und produktiver Konsumtion


```
c_g:=DM([3/2,21]):float(c_g);

c_r:=DM([15/2,29]):float(c_r);
```

```
c_r:=DM([15/2,29]):float(c_r);
c:=c_g+c_r;
```

$$\begin{pmatrix} 1.5 \\ 21.0 \end{pmatrix}$$

$$\begin{pmatrix} 7.5 \\ 29.0 \end{pmatrix}$$

$$\begin{pmatrix} 9 \\ 50 \end{pmatrix}$$

Allgemeine Formeln:

x=A*x+c	Spaltenvektor des Produktionsausstoßes nach Sektoren
$c=c_g+c_r+c_t$	Spaltenvektor der zur Konsumtion vorgesehenen Produkte, unterteilt nach Konsumtion der Lohnarbeiter, produktiver Konsumtion und Transfer
$x=(E-A)^{-1}*c$	Erforderlicher Produktionsausstoß bei vorgegebenem Konsumtionsziel
B'=B+(E-A)*Dg(x)	Bilanzmatrix (Spalten = Bestand der verschiedenen Güterarten pro Sektor)

Die Bilanz in der preislichen Darstellung

Preise für die einzelnen Güterarten:

```
p:=DM([[10,1]]);
```

$$\begin{pmatrix} 10 & 1 \end{pmatrix}$$

Löhne und Gewinne (zunächst in Arbeitsstunden wie bei Marx):

```
L:=DM([[20,16]]);
G:=DM([[50,54]]);
L+G;
```

$$\begin{pmatrix} 20 & 16 \end{pmatrix}$$

$$\begin{pmatrix} 50 & 54 \end{pmatrix}$$

$$\begin{pmatrix} 70 & 70 \end{pmatrix}$$

Organische Zusammensetzung der Kapitale in den beiden Sektoren:

```
v:=L*Dg(x)^(-1):float(v);

s:=G*Dg(x)^(-1):float(s);
```

```
s:=G*Dg(x)^(-1):float(s);
u:=DM([[G[i]/L[i]$i=1..2]]):float(u);
```

$$(2.0 \quad 0.16)$$

$$(5.0 \quad 0.54)$$

$$(2.5 \quad 3.375)$$

In der letzten Zeile findet sich die klassische sektorenspezifische Mehrwertrate *G(ewinn)/L(ohn)=s/v.*

Beachte, dass Lohn und Gewinn nicht mit der preislichen Bewertung der in den jeweilgen Sektoren produzierten Konsumgütern übereinstimmen; nur die Summen stimmen überein.

```
L<>p*Dg(c_g);
G<>p*Dg(c_r);
```

Lohn kauft also in der Summe alle Konsummittel der Lohnarbeiter aus, aber das gilt nicht sektorenweise. Z.B. könnte ein Sektor ausschließlich Vorprodukte produzieren.

Die summarische Gleichung folgt aus der Bilanzgleichung $g_a=p*c_a$ der Gegenüberstellung von Einküften und der preislichen Bewertung der Konsumgüter jedes einzelnen Marktteilnehmers bei einfacher Reproduktion.

Produktion mit verschiedenen Lohnarbeitsarten - Eine Rechnung ohne Unternehmer

Reinterpretation von *L* und *G* als Entlohnung von zwei Arten standardisierter Lohnarbeiten, die von zwei Lohnarbeiterfraktionen in beiden Produktionssektoren erbracht werden.

Zunächst die Arbeitsbilanz

```
hold(B*Dg(x))=DM([[20,16],[50,54]]);
B:=DM([[20,16],[50,54]])*Dg(x)^(-1):float(B);
y:=B*x;
```

x ist der Spaltenvektor des Produktionsausstoßes nach Güterarten (in für die jeweilige Güterart typischen Einheiten), *y* der Vektor der erforderlichen Arbeit nach Lohnarbeitsarten (in für die jeweilige Lohnarbeitsart typischen Einheiten).

Diesen Lohnarbeitsarten (Zeilen obiger Matrix) stehen kumulierte konsumtive Bedürfnisse der beiden Lohnarbeiterfraktionen gegenüber

```
hold(K=C*Dg(y))=float(concatMatrix(c_g,c_r));
C:=concatMatrix(c_g,c_r)*Dg(y)^(-1):float(C);
```

Wie sind Preise und Löhne festzusetzen, damit die Rechnung aufgeht?

Dafür ist zu bestimmen, wieviel für eine Einheit von Gut i zu bezahlen ist (Vektor $p=[p_1,p_2]$) und wieviel Lohn für eine Arbeitseinheit der Fraktion j auszuzahlen ist (Vektor $f=[f_1,f_2]$).

```
sys1:=[
// Lohn 1 kauft Konsum 1 aus
f_1*36=1.5*p_1+21*p_2,
// Lohn 2 kauft Konsum 2 aus
f_2*104=7.5*p_1+29*p_2];
```

In Matrixschreibweise: $f*Dg(y)=p*K=p*C*Dg(y)$ oder $f=p*C$.

Die Grundgleichung $p=p*A+v+s$ vereinfacht wegen $s=0$ zu $v=p*(E-A)$.
Da v, der Lohn L und die Matrix B über die Beziehung $f*B*Dg(x)=L=v*Dg(x)$ verbunden sind, gilt weiter $f*B=p*(E-A)$.

```
sys2:=DM([[f_1,f_2]])*B=DM([[p_1,p_2]])*(Id(2)-A);
```

Diese beiden Systeme ergeben ein homogenes lineares Gleichungssystem (vier Gleichungen in vier Unbestimmten), das in diesem Beispiel einen eindimensionalen Lösungsraum hat.

```
solve([op(sys1),(sys2[1][i]=sys2[2][i]$i=1..2)],[p_1,p_2,f_1,f_2]);
```

Die Preise p und Arbeitswertfaktoren f können also (bis auf skalare Faktoren) nicht anders gewählt werden als in Fleissners Beispiel von Anfang an postuliert.

f=p*C	Lohn und Konsum passen zueinander (globale Bedingung)
f*B=p*(E-A)	Lohn und Produktionserlös passen zueinander (lokale Bedingung)

Zwischenbilanz

Die Güter- und Arbeitsbilanzgleichungen $y=B*x$ und $x=A*x+c=A*x+C*y$ kann man zur Matrix

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} A & C \\ B & 0 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \end{pmatrix}$$

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} A & C \\ B & 0 \end{pmatrix} \cdot \begin{pmatrix} x \\ y \end{pmatrix}$$

zusammenfassen, die Bedingungen *f=p*C* und *p=p*A+f*B* für den Preisvektor *p* und den Vektor der Arbeitswertfaktoren *f* zur Matrix

$$\begin{pmatrix} p & f \end{pmatrix} = \begin{pmatrix} p & f \end{pmatrix} \cdot \begin{pmatrix} A & C \\ B & 0 \end{pmatrix}$$

In beiden Gleichungen spielt die Matrix

$$\mathbb{U} = \begin{pmatrix} A & C \\ B & 0 \end{pmatrix}$$

eine zentrale Rolle.

Aus beiden Gleichungen lässt sich *y* bzw. *f* eliminieren, woraus sich Bestimmungsgleichungen *x=(A+C*B)*x* und *p=p*(A+C*B)* für *x* und *p* sowie die zentrale Rolle der Matrix *U=A+C*B* ergeben.

Eine genauere Analyse zeigt, dass es sich dabei um Grenzgleichungen evolutionärer Iterationsverfahren handelt. Das ist der Kern der Fleissnerschen Argumentation zum Transformationsproblem.

```
DIGITS:=4:
p:=float(DM([[1,1]])):
l:=[p,((p:=p*(A+C*B))$i=1..6)]:
map(l,transpose);
delete DIGITS:
```

```
eigenvectors(transpose(A+C*B));
```

Der Kern der **These von der Arbeitskraft als Ware** stellt sich also wie folgt dar:

* Die These bezieht sich nicht auf Arbeitskraft schlechthin, sondern nur auf standardisierte Fertigkeiten, welche in verschiedenen Produktionen benötigt werden.

* Diese standardisierten, in verschiedenen Produktionen benötigten Fertigkeiten gehen in die Bilanzmatrix $\mathbb{U}$ auf dieselbe Weise ein wie die standardisierten, in verschiedenen Produktionen benötigten Vorprodukte.

* Im Rahmen eines Verdingungsverhältnisses wird die "Ware Arbeitskraft" durch Konsum der erforderlichen Lebensmittel im vorangegangenen Produktionstakt rechnerisch auf dieselbe Weise "produziert" wie andere Vorprodukte auch.

* Die Bewertungen *f* der verschiedenen Lohnarbeitsarten stehen in einem ähnlichen Konkurrenzverhältnis wie die Bewertungen *p* der verschiedenen Güterarten und sogar in Konkurrenz mit diesen.

* Jedoch sind die Vektoren *p* und *f* durch die elementare Beziehung $f=p*C$ miteinander verbunden, so dass **bei konstanten Preisen *p*** der Vektor *f* die Konkurrenz der einzelnen Lohnarbeitsarten um Anteile am Warenkorb $c_g=C*y$ der insgesamt für die Bedürfnisse von Lohnarbeit zur Verfügung stehenden Güter ausdrückt und damit **der gesellschaftliche Streit um die Definition dieser Bedürfnisse als Streit um Stücklohnnormen** geführt wird.

Wertrechnung unternehmerischer Tätigkeit

Das Grundszenario

Zwei Sektoren mit Lohnarbeiten und Arbeitswertfaktoren.

```
B_g:=DM([[7,0],[0,7/10]]); y_g:=B_g*x:
B_g*Dg(x); L;
f_g^0:=DM([[20/70,16/70]]);
f_g^0=DM([[v[i]/(v[i]+s[i])$i=1..2]]);
```

$$\begin{pmatrix} 7 & 0 \\ 0 & \frac{7}{10} \end{pmatrix}$$

$$\begin{pmatrix} 70 & 0 \\ 0 & 70 \end{pmatrix}$$

$$\begin{pmatrix} 20 & 16 \end{pmatrix}$$

$$\begin{pmatrix} \frac{2}{7} & \frac{8}{35} \end{pmatrix}$$

$$\begin{pmatrix} \frac{2}{7} & \frac{8}{35} \end{pmatrix} = \begin{pmatrix} \frac{2}{7} & \frac{8}{35} \end{pmatrix}$$

$$\left(\frac{2}{7}\ \frac{8}{35}\right) = \left(\frac{2}{7}\ \frac{8}{35}\right)$$

Setze konsumtive Bedürfnisse so, dass sie **robust gegen Änderungen der Preisproportionen *p* der Güter** sind. Dazu ist c_g entsprechend der y_g-Wichtung auf die Spalten von C_g zu verteilen - die Bedürfnisstrukturen der Arbeiter verschiedener Lohnarbeiterfraktionen werden als gleich angesetzt.

```
h:=f_g^0*y_g;
C_g:=h[1]^(-1)*c_g*f_g^0;
B_g;
f_g^0=p*C_g;
c_g=C_g*y_g;
L=p*C_g*y_g;
```

$$\left(36\right)$$

$$\begin{pmatrix} \frac{1}{84} & \frac{1}{105} \\ \frac{1}{6} & \frac{2}{15} \end{pmatrix}$$

$$\begin{pmatrix} 7 & 0 \\ 0 & \frac{7}{10} \end{pmatrix}$$

$$\left(\frac{2}{7}\ \frac{8}{35}\right) = \left(\frac{2}{7}\ \frac{8}{35}\right)$$

$$\begin{pmatrix} \frac{3}{2} \\ 21 \end{pmatrix} = \begin{pmatrix} \frac{3}{2} \\ 21 \end{pmatrix}$$

$$\left(20\ \ 16\right) = \left(36\right)$$

Das Erweiterungsszenario

Sollen die Unternehmer als **eine** weitere Fraktion **auf diesem fixierten Grundszenario** in die Rechnung einbezogen werden, so muss die Matrix C_g um eine Spalte c_r - die Bedürfnisse der Unternehmerfraktion - zur Matrix C und die Matrix B_g um eine Zeile b_r - die Arbeitsaufwendungen der Unternehmerfraktion - zur Matrix B ergänzt werden.

$y=B*x$ ist dabei durch einen weiteren Eintrag $y_r=b_r*x$ zu ergänzen, den Beitrag der Unternehmer zur Wertrechnung in UAE (unternehmerischen Arbeitseinheiten), die im "unternehmerischen Stundenzettel" $b_r*Dg(x)$ als Zeilenvektor sektorenweise wie in jeder Lohnarbeiterfraktion auch erfasst werden.

Da es nur auf die Proportionen in diesem Stundenzettel ankommt, postulieren wir $y_r=b_r*x=1$. Damit gibt $b_r*Dg(x)$ genau den Bruchteil des gesamten unternehmerischen Arbeitsaufwands an, welcher für die einzelnen Produktionen

erforderlich ist. Die Komponente *i* des Vektors b_r hat dann die Einheit $1/E_i$ und die erforderlich ist. Die Komponente *i* des Vektors b_r hat dann die Einheit $1/E_i$ und die Einheit UAE ist aus unseren weiteren Rechnungen verschwunden.

Als Grundgleichungen ergeben sich

$$U = U_g + c_r*b_r = A + C_g*B_g + c_r*b_r$$
$$x = (U_g + c_r*b_r)*x = A*x + c_g + c_r$$
$$p = p*(U_g + c_r*b_r) = p*A + p*(U_g - A) + p*c_r*b_r = p*A + v + s$$

und insbesondere für unser Beispiel

```
Ug:=A+Cg*Bg: float(Ug);
```

$$\begin{pmatrix} 0.1833333333 & 0.006666666667 \\ 3.166666667 & 0.3933333333 \end{pmatrix}$$

Variationspunkte in dieser Rechnung ist das Prinzip, nach dem der "unternehmerische Stundenzettel" ausgefüllt wird.

Beispiel 1: Stücklohn für Unternehmer

Annahme: Der unternehmerische Aufwand ist proportional zum Produktionsausstoß.

```
b:=DM([[1,1]]);
br:=b/(b*x)[1,1];
br*x;
U:=Ug+cr*br:float(U);
```

$$\begin{pmatrix} 1 & 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{110} & \frac{1}{110} \end{pmatrix}$$

$$\begin{pmatrix} 1 \end{pmatrix}$$

$$\begin{pmatrix} 0.2515151515 & 0.07484848485 \\ 3.43030303 & 0.656969697 \end{pmatrix}$$

Mögliche Preisproportionen ergeben sich als Lösung der Gleichung $p=p*U$:

```
sys:=U-Id(2);
p1:=DM([[-sys[2,1]/sys[1,1],1]]);
float(p1);
```

$$\begin{pmatrix} -\frac{247}{330} & \frac{247}{3300} \\ \frac{566}{165} & -\frac{283}{825} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1132}{247} & 1 \end{pmatrix}$$

$$\begin{pmatrix} 4.582995951 & 1.0 \end{pmatrix}$$

$$(4.582995951 \quad 1.0)$$

Die wertmäßige Beurteilung in der Marxschen Notation durch Stückpreis, vorgeschossenes konstantes Kapital *K=p*A*Dg(x)*, Lohn *L=v*Dg(x)*, Gewinn *G=s*Dg(x)* sowie Mehrwertrate *G/L* und Profitrate *G/(K+L)* für die einzelnen Produktionssektoren lässt sich nun wie folgt berechnen:

```
displayResult:=proc(u) local Ausgabe,uhu;
begin
Ausgabe:=["Stückpreis","konstantes Kapital","Lohn",
"Gewinn","Mehrwertrate","Profitrate"];
uhu:=zip(Ausgabe,float(u),
  (a,b)-> _concat(stringlib::format(a,20,Center),
    stringlib::formatf(b[1],3,15), stringlib::formatf(b[2],3,15)));
print(NoNL,"\n".stringlib::subs(
  output::tableForm(uhu,String,Sort=FALSE),"\""=""))
end_proc:
```

```
p:=p1:
cc:=p*A: v:=p*(U_g-A): s:=p*c_r*b_r:
mr:=DM([[s[i]/v[i]$i=1..2]]):
pr:=DM([[s[i]/(v[i]+cc[i])$i=1..2]]):
displayResult([p,cc*Dg(x),v*Dg(x),s*Dg(x),mr,pr]):
```

Stückpreis	4.583	1.000
konstantes Kapital	24.583	30.000
Lohn	15.486	12.389
Gewinn	31.686	31.686
Mehrwertrate	2.046	2.558
Profitrate	0.791	0.748

Probe: Lohn und Gewinn kaufen genau die bestellten Konsumgüter aus

```
[v*x=p*c_g,s*x=p*c_r]
```

$$\left[\left(\frac{6885}{247} \right) = \left(\frac{6885}{247} \right), \left(\frac{15653}{247} \right) = \left(\frac{15653}{247} \right)\right]$$

Preisbestimmung durch iterativen Ansatz wie bei Fleissner:


```
DIGITS:=5:
p:=float(DM([[1,1]])):
l:=[p,(p:=p*U)$i=1..5]:
map(l,transpose);
```

```
map(l,transpose);
map(l,u->float(u[1]/u[2]));
delete DIGITS;
```

$$\left[\begin{pmatrix} 1.0 \\ 1.0 \end{pmatrix}, \begin{pmatrix} 3.6818 \\ 0.73182 \end{pmatrix}, \begin{pmatrix} 3.4364 \\ 0.75636 \end{pmatrix}, \begin{pmatrix} 3.4589 \\ 0.75411 \end{pmatrix}, \begin{pmatrix} 3.4568 \\ 0.75432 \end{pmatrix}, \begin{pmatrix} 3.457 \\ 0.7543 \end{pmatrix}\right]$$

$$[1.0,\ 5.0311,\ 4.5433,\ 4.5866,\ 4.5827,\ 4.583]$$

Beispiel 2: Erfolgsunabhängige Entlohnung

Annahme: Der unternehmerische Arbeitsaufwand ist unabhängig vom Produktionsausstoß

Ansatz:

```
b:=DM([[1,1]])*Dg(x)^(-1);
b_r:=b/(b*x)[1,1];
b_r*Dg(x);
U:=U_g+c_r*b_r:float(U);
```

$$\begin{pmatrix} \frac{1}{10} & \frac{1}{100} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{20} & \frac{1}{200} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} & \frac{1}{2} \end{pmatrix}$$

$$\begin{pmatrix} 0.5583333333 & 0.04416666667 \\ 4.616666667 & 0.5383333333 \end{pmatrix}$$

Lösung des Gleichungssystems:

```
sys:=U-Id(2);
p2:=DM([[-sys[2,1]/sys[1,1],1]]);
float(p2);
```

$$\begin{pmatrix} -\frac{53}{120} & \frac{53}{1200} \\ \frac{277}{60} & -\frac{277}{600} \end{pmatrix}$$

$$\begin{pmatrix} \frac{554}{53} & 1 \end{pmatrix}$$

$$\begin{pmatrix} 10.45283019 & 1.0 \end{pmatrix}$$

Auswertung:

```
p:=p2:
cc:=p*A: v:=p*(U_g-A): s:=p*c_r*b_r:
mr:=DM([[s[i]/v[i]$i=1..2]]):
pr:=DM([[s[i]/(v[i]+cc[i])$i=1..2]]):
displayResult([p,cc*Dg(x),v*Dg(x),s*Dg(x),mr,pr]):
```

Stückpreis	10.453	1.000
konstantes Kapital	30.453	30.000
Lohn	20.377	16.302
Gewinn	53.698	53.698
Mehrwertrate	2.635	3.294
Profitrate	1.056	1.160

Beispiel 3: Gleiche Mehrwertrate

Annahme: Der unternehmerische Gewinn führt zu gleichen Mehrwertraten.

Mit $b_r \sim v = f_g * B_g$ ist über f_g in diesem Fall eine der "Zutaten" zur Berechnung des Preisvektors bereits abhängig von der preislichen Bewertung der "in Bewegung gesetzten Agentien", hier der Lohnarbeiten. Da die Lohnarbeiten zum Grundszenario gehören, diese Arbeitswertfaktoren also hier als konstant vorausgesetzt sind, **bleibt das System linear.**

```
b:=fg0*Bg;
br:=b/(b*x)[1,1];
br*Dg(x);
U:=Ug+cr*br:float(U);
```

$$\left(2 \quad \frac{4}{25} \right)$$

$$\left(\frac{1}{18} \quad \frac{1}{225} \right)$$

$$\left(\frac{5}{9} \quad \frac{4}{9} \right)$$

$$\begin{pmatrix} 0.6 & 0.04 \\ 4.777777778 & 0.5222222222 \end{pmatrix}$$

Lösung des Gleichungssystems:

```
sys:=U-Id(2);
p3:=DM([[-sys[2,1]/sys[1,1],1]]);
float(p3);
```

$$\begin{pmatrix} -\frac{2}{5} & \frac{1}{25} \\ \frac{43}{9} & -\frac{43}{90} \end{pmatrix}$$

$$\left(\frac{215}{18} \quad 1 \right)$$

$$\left(11.94444444 \quad 1.0 \right)$$

Auswertung:

```
p:=p3:
```

```
p:=p3:
cc:=p*A: v:=p*(U_g-A): s:=p*c_r*b_r:
mr:=DM([[s[i]/v[i]$i=1..2]]):
pr:=DM([[s[i]/(v[i]+cc[i])$i=1..2]]):
displayResult([p,cc*Dg(x),v*Dg(x),s*Dg(x),mr,pr]):
```

Stückpreis	11.944	1.000
konstantes Kapital	31.944	30.000
Lohn	21.620	17.296
Gewinn	65.880	52.704
Mehrwertrate	3.047	3.047
Profitrate	1.230	1.114

Beispiel 4: Gleiche Profitrate

Annahme: Der unternehmerische Gewinn führt zu gleichen Profitraten.

Mit $b_r \sim p*A+f_g*B_g$ geht nun der zu bestimmende Preisvektor in die Einträge der Matrix U selbst ein. Der "unternehmerische Stundenzettel" hängt also von der preislichen Bewertung der "in Bewegung gesetzten Agentien" ab. **Das Gleichungssystem wird nichtlinear.**

Wir führen deshalb die bisherigen Rechnungen symbolisch mit einem Preisvektor p=[t 1] aus und extrahieren aus dem Ergebnis die Bedingungen zur Bestimmung von t.

Ansatz:

```
p:=DM([[t,1]]);
b:=p*U_g;
b_r:=normal(b/(b*x)[1,1]);
U:=normal(U_g+c_r*b_r);
```

$$\begin{pmatrix} t & 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{11 \cdot t}{60} + \frac{19}{6} & \frac{t}{150} + \frac{59}{150} \end{pmatrix}$$

$$\begin{pmatrix} \frac{11 \cdot t + 190}{150 \cdot t + 4260} & \frac{t + 59}{375 \cdot t + 10650} \end{pmatrix}$$

$$\begin{pmatrix} \frac{55 \cdot t + 1103}{75 \cdot t + 2130} & \frac{20 \cdot t + 1027}{750 \cdot t + 21300} \\ \frac{397 \cdot t + 9500}{75 \cdot t + 2130} & \frac{353 \cdot t + 11800}{750 \cdot t + 21300} \end{pmatrix}$$

Lösung des Gleichungssystems:


```
sys:=U-Id(2);
q:=normal(DM([[-sys[2,1]/sys[1,1],1]]));
```

$$\begin{pmatrix} \frac{55 \cdot t + 1103}{75 \cdot t + 2130} - 1 & \frac{20 \cdot t + 1027}{750 \cdot t + 21300} \\ \frac{397 \cdot t + 9500}{75 \cdot t + 2130} & \frac{353 \cdot t + 11800}{750 \cdot t + 21300} - 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{55 \cdot t+1103}{75 \cdot t+2130} - 1 & \frac{20 \cdot t+1027}{750 \cdot t+21300} \\ \frac{397 \cdot t+9500}{75 \cdot t+2130} & \frac{353 \cdot t+11800}{750 \cdot t+21300} - 1 \end{pmatrix}$$

$$\left(\frac{397 \cdot t+9500}{20 \cdot t+1027} \quad 1 \right)$$

Nun können aus *p(t)=q(t)* die möglichen Lösungen *t* bestimmt werden, was in unserem Fall auf eine quadratische Gleichung mit zwei Lösungen führt:

```
sol:=solve(p[1]=q[1],[t]);
```

$$\left\{ \left[t = -\frac{\sqrt{11569}}{4} - \frac{63}{4} \right], \left[t = \frac{\sqrt{11569}}{4} - \frac{63}{4} \right] \right\}$$

Von den beiden Lösungen ist allein die positive relevant.

```
p4:=subs(p,sol[2]);float(p4);
```

$$\left(\frac{\sqrt{11569}}{4} - \frac{63}{4} \quad 1 \right)$$

$$(11.13982149 \quad 1.0)$$

Auswertung - beachte, dass im Ansatz auch $b_r=b_r(t)$ eine Funktion von *t* war.

```
p:=p4:
br:=subs(br,sol[2]): float(br);
cc:=p*A: v:=p*(Ug-A): s:=p*cr*br:
mr:=DM([[s[i]/v[i]$i=1..2]]):
pr:=DM([[s[i]/(v[i]+cc[i])$i=1..2]]):
displayResult([p,cc*Dg(x),v*Dg(x),s*Dg(x),mr,pr]):
```

$$(0.05269591088 \quad 0.004730408912)$$

Stückpreis	11.140	1.000
konstantes Kapital	31.140	30.000
Lohn	20.950	16.760
Gewinn	59.309	53.240
Mehrwertrate	2.831	3.177
Profitrate	1.139	1.139

Für eine iterative Lösung ist zu berücksichtigen, dass die Transformationsmatrix $U=U(p)=U_g+c_r*b_r$ über $b_r=b_r(p)$ selbst auch von *p* abhängt, so dass in jedem Schritt $p'=p*U(p)$ zu berechnen ist.

Dazu definieren wir zunächst eine Funktion, die $U=U(p)$ aus dem aktuellen Preisvektor berechnet.

berechnet.

```
U_proc:=proc(p) local b,b_r;
begin b:=p*U_g; b_r:=(b/(b*x)[1,1]); (U_g+c_r*b_r) end_proc;
```

$\text{proc } U_{\text{proc(p)}} \ldots \text{ end}$

Und nun die Iteration:

```
DIGITS:=5:
p:=float(DM([[1,1]])):
l:=[p,(p:=p*U_proc(p))$i=1..6]:
map(l,transpose);
map(l,u->float(u[1]/u[2]));
delete DIGITS;
```

$$\left[\begin{pmatrix} 1.0 \\ 1.0 \end{pmatrix}, \begin{pmatrix} 5.0136 \\ 0.59864 \end{pmatrix}, \begin{pmatrix} 5.6259 \\ 0.53741 \end{pmatrix}, \begin{pmatrix} 5.7573 \\ 0.52427 \end{pmatrix}, \begin{pmatrix} 5.7874 \\ 0.52126 \end{pmatrix}, \begin{pmatrix} 5.7944 \\ 0.52056 \end{pmatrix}, \begin{pmatrix} 5.7961 \\ 0.52039 \end{pmatrix}\right]$$

$$[1.0,\ 8.375,\ 10.468,\ 10.982,\ 11.103,\ 11.131,\ 11.138]$$

Ein Blick auf die Eigenwerte der (symbolischen) Matrix U:

```
eigenvalues(U);
```

$$\left\{1, \frac{51 \cdot t + 510}{250 \cdot t + 7100}\right\}$$

Zusammenfassung

(1) In der vertikalen Gliederung nach Produktionssektoren kommen nicht nur standardisierte Vorprodukte (Matrix *A* der technischen Koeffizienten), sondern auch standardisierte Arbeiten (Matrix *B* der Arbeitsaufwandskoeffizienten) zum Einsatz. Die Masse der Lohnarbeiter zerfällt auf diese Weise horizontal in verschiedene Fraktionen mit vergleichbaren Fähigkeiten und organischer Zusammensetzung der konsumtiven Bedürfnisse. In dem Sinne kann der Wertbeitrag der Unternehmer quantitativ als spezielle Fraktion der Lohnarbeiter behandelt werden.

(2) Ich unterscheide zwischen Arbeitsaufwand und Arbeitswert. Der Arbeitsaufwand ist ein Verhältnis *innerhalb* einer speziellen Fraktion i der Lohnarbeiter und misst deren reales Arbeitsergebnis in für die jeweilige Lohnarbeiterfraktion typischen Arbeitseinheiten AE_i (Anzahl von Stanzstücken, Kisten gepflückter Äpfel, Anwesenheitsstunden etc.). Der Arbeitswert ergibt sich daraus durch Multiplikation mit dem für die jeweilige Lohnarbeiterfraktion spezifischen Arbeitswertfaktor f_i (Einheit: GE/AE_i). In diesem Sinne ist *jede* Arbeit multiplizierte Arbeit. Dies ist der **Kern von Wert als gesellschaftlichem Verhältnis** und nicht zu vernachlässigen, um sich "die Mühe der Reduktion" (MEW23, S. 59) zu ersparen.

(3) Das Wertverhältnis als gesellschaftliches Verhältnis ist in der Höhe dieser Arbeitswertfaktoren kodiert, welche die Höhe des Anteils der jeweiligen Lohnarbeiterfraktion am konsumtiven Gesamtprodukt bestimmen. Das Ringen der einzelnen Lohnarbeiterfraktionen um ihren Anteil am gesellschaftlichen Produkt findet also im Kampf um die Höhe von Stücklohnnormen ihren Ausdruck. Der Juni 1953 lässt grüßen.

(4) Die durchschnittliche Profitrate ergibt sich in dieser Rechnung als Arbeitswertfaktor der Fraktion der Unternehmer auf der Basis einer spezifischen Wertform, eine (hypothetische) durchschnittlich gleiche Mehrwertrate als Arbeitswertfaktor auf der Basis einer anderen Wertform, ein äquivalenzökonomischer Kalkül als Arbeitswertfaktor auf der Basis einer dritten Wertform. Die Möglichkeit der Erweiterung des Modells auf mehrere Unternehmerfraktionen ist offensichtlich.

Anhang: Code zum Erzeugen gewisser Formeln


```
DM([[hold(x)],[hold(y)]])=hold(_mult)
(DM([[hold(A),hold(C)],[hold(B),0]]),DM([[hold(x)],[hold(y)]]));
```

```
DM([[hold(p),hold(f)]])=hold(_mult)
(DM([[hold(p),hold(f)]]),DM([[hold(A),hold(C)],[hold(B),0]]))
```

```
Symbol::Uopf=DM([[hold(A),hold(C)],[hold(B),0]])
```

```
Symbol::sim
```

~

```
Symbol::Uopf
```

𝕌, ≈